**********************************
**********************************
***Base de observación de clase***
**********************************
**********************************

clear
set more off

import excel "$basein/Instrumentos valora/Base_observacion_limpia.xlsx"

drop CO CP CQ CR CS CT CU CV CW

drop in 1/2

foreach x in AC	AD	AE	AF	AG	AH	AI	AJ	AK	AL	AM	AN	AO	AP	AQ	AR	AS	AT	AU	AV	AW	AX	AY	AZ	BA	BB	BC	BD	BE	BF	BG	BH	BI	BJ	BK	BL	BM	BN	BO	BP	BQ	BR	BS	BT	BU	BV	BW	BX	BY	BZ	CA	CB {

	gen `x'_num=substr(`x',1,1)

	destring `x'_num, replace force

}

***Características de la implementación***

egen Indice_1=rowmean(AC_num-AN_num)
egen Indice_2=rowmean(AO_num-AX_num)
egen Indice_3=rowmean(AY_num-BP_num)
egen Indice_4=rowmean(BQ_num-CB_num)
egen Indice_total=rowmean(Indice_*)

***Uso de los aprendizajes***

foreach x in CD CE CF {
	replace `x'="" if `x'=="5Mucho"
}


multencode CD CE CF , generate(uSO_CD uSO_CE uSO_CF)

***Genero variables con escala de uno a 4***

gen CD_final=.
replace CD_final=0 if uSO_CD==6
replace CD_final=3 if uSO_CD==1
replace CD_final=6 if uSO_CD==7
replace CD_final=12 if uSO_CD==8
replace CD_final=20 if uSO_CD==4

gen CE_final=.
replace CE_final=0 if uSO_CE==6
replace CE_final=3 if uSO_CE==1
replace CE_final=6 if uSO_CE==2
replace CE_final=12 if uSO_CE==3
replace CE_final=20 if uSO_CE==4

gen CF_final=.
replace CF_final=0 if uSO_CF==5
replace CF_final=3 if uSO_CF==1
replace CF_final=6 if uSO_CF==7
replace CF_final=12 if uSO_CF==3
replace CF_final=20 if uSO_CF==4

egen Indice_uso=rowmean(*_final)

tabstat Indice_*, by(L)

rename J cct
keep cct Indice_* 

save "$base_out/Observacion_indices.dta", replace 								/***Save base***/

merge 1:1 cct using "$basein/Datos adicionales/Tratamientos_completos.dta"

keep if _merge==2 | _merge==3

***Traigo variables de DWMS***

rename cct school_id

merge 1:1 school_id using "$base_out/DWMS_Regresiones", gen(_DWMS)

drop if secondary==1

******************
***KEEP´ONLY PEC VS PEC PLUS******
******************

drop if state==10
drop if treatgroup2=="Control"
drop if treatgroup2==""
**Estsandarizo indicadores de implementacion y uso***

foreach x in Indice_1 Indice_2 Indice_3 Indice_4 Indice_total Indice_uso {

egen `x'_std=std(`x')

}

***Hago tablas de regresiones***
label var Indice_total_std "Implementation index Stallings"
label var Indice_uso_std "Use index Stallings"


eststo clear
foreach x in Indice_total Indice_uso {
	eststo valora1_`x': reghdfe management_sted3 `x'_std, a(i.estrato i.enumerador) vce(cluster school_id)
}



**************************
***Base de herramientas***
**************************

clear
set more off


import excel "$basein/Instrumentos valora/Base_herramientas_limpia.xlsx"


**Elimino primeras filas que configuracon los labels

drop in 1/2

foreach x in Y	Z	AA	AB	AC	AD	AE	AF	AG	AH	AI	AJ	AK	AL	AM	AN	AO	AP	AQ	AR	AS	AT	AU	AV	AW	AX	AY	AZ	BA	BB	BC	BD	BE	BF	BG	BH	BI	BJ	BK	BL	BM	BN	BO	BP	BQ  {

	gen `x'_num=substr(`x',1,1)

	destring `x'_num, replace force

}

***Características de la implementación***

egen IndiceH_1=rowmean(Y_num-AF_num)
egen IndiceH_2=rowmean(AG_num-AN_num)
egen IndiceH_3=rowmean(AO_num-AV_num)
egen IndiceH_4=rowmean(AW_num-BB_num)
egen IndiceH_5=rowmean(BC_num-BJ_num)
egen IndiceH_6=rowmean(BK_num-BQ_num)

egen IndiceH_total=rowmean(IndiceH_*)


**Uso de las herramientas**

foreach x in BT BV BX {

	tab `x'
	replace `x'="" if `x'=="4Algo"
	replace `x'="" if `x'=="2Casi nada"

	gen `x'_final=.
	replace `x'_final=0 if `x'=="A ninguno"
	replace `x'_final=5 if `x'=="Menos de 10%"
	replace `x'_final=12.5 if `x'=="Entre 10 y 25%"
	replace `x'_final=38 if `x'=="Entre 26 y 50%"
	replace `x'_final=62.5 if `x'=="Entre 51 y 75%"
	replace `x'_final=88 if `x'=="Entre 76 y 100%"

}

egen IndiceH_uso=rowmean(*_final)

tabstat IndiceH_*, by(L)

rename J cct
keep cct IndiceH_*

save "$base_out/Herramientas_indices.dta", replace

merge 1:1 cct using "$basein/Datos adicionales/Tratamientos_completos.dta"

keep if _merge==2 | _merge==3

***Traigo variables de DWMS***

rename cct school_id

merge 1:1 school_id using "$base_out/DWMS_Regresiones", gen(_DWMS)

drop if secondary==1

******************
***KEEP´ONLY PEC VS PEC PLUS******
******************

drop if state==10
drop if treatgroup2=="Control"


**Estsandarizo indicadores de implementacion y uso***

foreach x in IndiceH_1 IndiceH_2 IndiceH_3 IndiceH_4 IndiceH_5 IndiceH_6 IndiceH_total IndiceH_uso {

egen `x'_std=std(`x')

}

***Hago tablas de regresiones***
label var IndiceH_total_std "Implementation index SisAT"
label var IndiceH_uso_std "Use index SisAT"

foreach x in IndiceH_total IndiceH_uso  {
	eststo valora1_`x': reghdfe management_sted3 `x'_std, a(i.estrato i.enumerador) vce(cluster school_id)
}

esttab valora1_Indice_total valora1_Indice_uso valora1_IndiceH_total valora1_IndiceH_uso  using "$tables/valora_1a.tex" ,  ///
fragment se ar2   b(a2) se(a2) nocon nonumber nomtitles nolines nogaps /// 
replace collabels(none) ///
coeflabels(Indice_total_std "Implementation index Stallings" Indice_uso_std "Use index Stallings" IndiceH_total_std "Implementation index SisAT" IndiceH_uso_std "Use index SisAT")  ///
keep(*_std) stats(N , fmt("%9.2gc") labels("No. of obs." )) star(* 0.10 ** 0.05 *** 0.01)
	
	



*****@@@@@@@@@@@@@@@@@@@*****
*****************************
*****Resultados Planea*******
*****************************
*****@@@@@@@@@@@@@@@@@@@*****

clear
set more off

use "$basein/Planea/alum_pb_2018"

drop if entidad=="CHIAPAS"

rename cct school_id

*********************************************************
***Traigo base con variable de tratamiento y controles***
*********************************************************

merge m:1 school_id using "$basein/Datos adicionales/base_Contraste.dta", gen(planea)

keep if planea==2 | planea==3


merge m:1 school_id using "$basein/Datos adicionales/Experimento_id.dta", gen(experimento)
drop experimento

merge m:1 school_id using "$basein/Datos adicionales/estratos.dta", gen(_estrato)

keep if _estrato==3

merge m:1 school_id using "$basein/Datos adicionales/Change_Director_2015_2018.dta", gen(_cambio)

drop if _cambio==2

******************
***KEEP´ONLY PEC VS PEC PLUS******
******************

drop if state==10
drop if treatgroup2=="Control"

*************
**Id estado**

gen esi=substr(school_id, 1,2)

**********************
**Genero tratamiento**

gen PEC_Plus=0
replace PEC_Plus=1 if treatgroup2=="PEC Plus"

gen PEC=0
replace PEC=1 if treatgroup2=="PEC"

gen Control=0
replace Control=1 if treatgroup2=="Control"

******************************************
**Estudiantes que no tomaron son missing**

replace wle_mat=. if presen_mat=="0"
replace wle_lyc=. if presen_esp=="0"

***********************************
**Estandarizo variables resultado**

egen mat_std=std(wle_mat)
egen leg_std=std(wle_lyc)
gen promedio=(wle_mat+wle_lyc)/2
egen promedio_std=std(promedio)

pca wle_mat wle_lyc, components(1)

predict prom_pca


**************************************************
**Genero efecto fijo de estado y otros controles**

encode entidad, gen(estado)

gen m_alta=0
replace m_alta=1 if marginacion=="Muy alta o Alta"

gen m_media=0
replace m_media=1 if marginacion=="Media"

gen m_baja=0
replace m_baja=1 if marginacion=="Baja o Muy baja"

encode school_id, gen(cct_num)

save "$base_out/PLANEA_temp", replace

rename school_id cct

**Pegados**

merge m:1 cct using "$base_out/Observacion_indices.dta", gen(_obs)

merge m:1 cct using "$base_out/Herramientas_indices.dta", gen(_herr)

keep if _obs==3 | _herr==3

**Estsandarizo indicadores de implementacion y uso***

foreach x in Indice_total Indice_uso IndiceH_total IndiceH_uso {

egen `x'_std=std(`x')

}


eststo clear
foreach var in mat leg promedio{
foreach x in Indice_total Indice_uso IndiceH_total IndiceH_uso {
	eststo: reghdfe `var'_std `x'_std, a(i.estrato ) vce(cluster cct)
	}
}

esttab  using "$tables/valora_2a.tex" ,  ///
fragment se ar2   b(a2) se(a2) nocon nonumber nomtitles nolines nogaps /// 
replace collabels(none) ///
coeflabels(Indice_total_std "Implementation index Stallings" Indice_uso_std "Use index Stallings" IndiceH_total_std "Implementation index SisAT" IndiceH_uso_std "Use index SisAT")  ///
keep(*_std) stats(N , fmt("%9.2gc") labels("No. of obs." )) star(* 0.10 ** 0.05 *** 0.01)
	
	
